

Solution to Exercise 2: Variability in serial smear results

Key Learning Points

When you have a hypothesis to test, remember that it may be logical to:

- Create and use a subset of the working dataset
- Create new variable(s)

Tasks:

Exercise hypothesis:

H₀: In each study country, at least 60% of cases found among suspects with a complete diagnostic series show a variation in the serial pattern

- Determine with a program C_EX02.PGM the proportions of smears with and without variation in serial smears by country*
- Interpret the findings*

Solution

Determine with a program C_EX02.PGM the proportions of smears with and without variation in serial smears by country

The following output was created:

Grading variation						
Study country	No variation	%	With variation	%	Total	%
Moldova	318	{36.6}	552	{63.4}	870	{100.0}
Mongolia	957	{63.8}	542	{36.2}	1499	{100.0}
Uganda	1857	{53.6}	1608	{46.4}	3465	{100.0}
Zimbabwe	1302	{63.0}	764	{37.0}	2066	{100.0}
Total	4434	{56.1}	3466	{43.9}	7900	

Percents: {Row}

Solution

Interpret the findings

The analysis with 95% confidence intervals for each individual country gave the following (see output next page).

Conclusion: Except for Moldova, the hypothesis has to be refuted for each country. Of course, there is no accepted standard what constitutes an “acceptable” minimum level of variation that should be found. Nevertheless, it would appear that the level of variation particularly in Mongolia and Zimbabwe is unexpectedly low, that is the serial results raise some questions on the diligence of reading and reporting sputum smear examination results.

"Moldova"

Grading variation			
	N	%	(95% CI)
No variation	318	36.6	(33.4-39.8)
With variation	552	63.4	(60.2-66.6)
Total	870	100.0	

"Mongolia"

Grading variation			
	N	%	(95% CI)
No variation	957	63.8	(61.4-66.2)
With variation	542	36.2	(33.8-38.6)
Total	1499	100.0	

"Uganda"

Grading variation			
	N	%	(95% CI)
No variation	1857	53.6	(51.9-55.2)
With variation	1608	46.4	(44.8-48.1)
Total	3465	100.0	

"Zimbabwe"

Grading variation			
	N	%	(95% CI)
No variation	1302	63.0	(60.9-65.1)
With variation	764	37.0	(34.9-39.1)
Total	2066	100.0	

The program C_EX02.PGM that produced the above output is the following:

```
* Program name: c_ex02.pgm
* Identifying patterns of serial smear results with identical individual results
* Objective of the exercise
* Identify series of identical result patterns in the four countries
* The reason for this exercise is that we hypothesize
*   that too regular patterns indicate that the laboratory
*   simply copies a positive result once found to (a) subsequent
*   result(s) rather than properly examining the individual smear
* Thus, this analysis may be an indirect quality assurance program
* First decision: denominator:
* Define the denominator with the choice of the appropriate dataset
*   Data set must be suspects
*   Assessing variability among persons with only negative results
*   is biased as the proportion of these varies widely, thus excluding
*   such examinees
*   Assessing variability among patients with only two results provides
*   too little insight in variability, selecting thus those with three
*   results of which at least one is positive
*   Furthermore, those with unquantified positive results will also
*   bias the result
cd c:\epidata_course
cls
logclose
close
```

```

read "c_ex01.rec"

* include only suspects for analysis
select reason=0

cls
* Select only examinees with three quantified smear results
select result1<4
select result2<4
select result3<4
* 61,064 records retained

* Select only suspects with at least 1 non-negative result
define allneg #
allneg=1
if (result1=0) and (result2=0) and (result3=0) then allneg=0
select allneg=1
* 7,900 records retained

savedata "temp.rec" /replace

*****
cls
close

read "temp.rec"

define variation #
variation=1
if (result1=result2) and (result1=result3) then variation=0
label variation "Grading variation"
labelvalue variation /0="No variation"
labelvalue variation /1="With variation"

cls
set echo=off
tables variation country /r
select country=1
Title "Moldova"
freq variation /c /ci
select
select country=2
Title "Mongolia"
freq variation /c /ci
select
select country=3
Title "Uganda"
freq variation /c /ci
select
select country=4
Title "Zimbabwe"
freq variation /c /ci
select
set echo=on

*****
* Clean up

close
erase "temp.chk"
erase "temp.rec"

```